

PROGRAMAS SOBRE CURVAS PLANAS

Por RAMON MARTUL ALVAREZ DE NEYRA
Profesor de la E.T.S.A. de La Coruña

Es realmente difícil encontrar un campo de las Matemáticas que haya sido tratado con mayor cariño, durante siglos, que el de la construcción de curvas. Basta observar sus nombres: la espiral de Arquímedes, el caracol de Pascal, el folium de Descartes, los óvalos de Cassini, la campana de Gauss, las curvas de Pearson, las de Lisşajous y un enorme etcétera, así como curvas que nos han llegado con sus nombres propios, sonoros, que tienen viejas resonancias griegas o latinas como cicloides, epicicloides, hipocicloides, tractrices, conoides, cisoides o también nombres expresivos, casi definiciones, como cardioides, rosas, lazos, cónicas, catenarias, trisectrices y muchas más. Curvas útiles que responden a la vieja definición: El conjunto de los puntos del plano que verifican tal o cual propiedad, y que han sido usadas para analizar el comportamiento conjunto de dos variables o, como los ábacos, para medir; auténticas calculadoras de años no tan lejanos. Curvas que intervienen en casi todos los campos del conocimiento y la investigación, como los polinomios de Bernstein, que sirven de base a diversos splines profusamente utilizados en el diseño asistido por ordenador y han sido precisamente los ordenadores los que han demostrado una mayor eficacia como instrumento trazador, a través de impresoras, digitalizadores y plotters, permitiendo representaciones de gran precisión y con un parámetro fundamental: la rapidez.

$$x = \sum_{i=1}^M R_i(t) \operatorname{sen} A_i(t)$$

$$y = \sum_{i=1}^M F_i(t) \operatorname{sen} B_i(t) \text{ con } R_i(t), F_i(t), A_i(t), B_i(t) \text{ e } \mathbb{R}(t)$$



y de éstas en el ínfimo subconjunto:

$$x = R \cdot \cos(t) + F \cdot \cos(D \cdot t)$$

$$y = R \cdot \operatorname{sen}(t) + J \cdot \operatorname{sen}(L \cdot t) \text{ con } R, F, Y, D, L \text{ e } \mathbb{R}$$

Podría haber sido cualquier otro. Aún así, se trata de una macrofamilia, pues en ella tienen cabida muchas de las citadas anteriormente.

Escogiendo convenientemente el intervalo de variabilidad de «t» y dando a los parámetros valores racionales, son siempre cerradas. Son curvas de gran belleza, simétricas respecto al eje OX, en función de t, y en muchos casos también al OY. Convenientemente manipuladas, se puede conseguir con ellas



Pues bien, este parámetro no es precisamente despreciable, sobre todo si pensamos que el cardinal infinito de las curvas del plano es infinitamente mayor que el cardinal infinito de los números reales, Aleph-sub2 contra Aleph-sub1, y que además si, como en muchos casos que veremos a continuación, tienen treinta y tantos o más puntos de máximo, mínimo o inflexión, representan un trabajo largo y tedioso.

De todas las curvas, fijémonos en las parametrizables y dentro de ellas en las que responden a unas ecuaciones del tipo:

la impresión de superficies regladas en perspectiva central, o cintas, o nada: simplemente el dibujo por el dibujo.

Los programas que permiten su representación se encuentran en el fichero de Matemáticas del Aula de Informática de la E.T.S.A. de La Coruña, con los nombres de «Curvas 1», «Curvas 2» y «Curvas 3». Llevan todos ellos factores de escala en los dos ejes, pero si $|RI + IFI| \approx 8$ y $|RI + IJI| \approx 6$ con factor 1 en ambos, ni se desmarcan ni son pequeñas. Es conveniente que los parámetros D y L no sean demasiado grandes ≤ 10 ,



pues en ese caso, las curvas son excesivamente complejas. Estas normas no son absolutas.

El programa «Curvas 1» permite la creación, en un principio aleatoria, de la curva periférica que es el soporte más adecuado para la elección del punto inicial o el de fuga o del tamaño de la «cinta» y es un archivo de datos imprescindible en los demás programas. Puede ser conveniente la utilización, en él, de un STEP alto, 0.1 aprox., para acelerar la ejecución. Estos programas han sido diseñados para un micro HP sobre plotter.

A modo de guía, para las primeras utilizaciones de estos programas, se proponen a continuación diversas combinaciones de valores de los parámetros que permiten hacerse una idea aproximada de las posibilidades de las curvas de esta familia.

Deben ser realizadas sobre «Curvas 1» y modificar en cada caso los factores de escala en los ejes, hasta que se adapten a lo que de ellas queramos conseguir.

R	6	4	4	4	4	4	4	4	4	4	16	14.5	4	6	4	4	4	4	4
F	-4.5	-1	-1	1	1	-5	3	-3	-3	-3	-12	-3.6	3	-4	-2	3	-3	4	3
D	2	-8	-8	-8	8	3	5	2	4	8/3	29/9	-5	3	4	5	5	-3	-3	
L	6	-8	-8	-8	8	1.5	6	3	-3	8/3	29/9	6	-2	-3	5	-2	2.5	4	
J	-4	-2	-1	-1	1	2	2	-2	-3	-12	-3.6	2	1.5	3	3	2	2.5	3	
RO	2	2	2	2	2	2	2	2	2	6	18	2	2	2	2	2	2	2	2
ESCALA X/Y	1/1	1/1	1/1	1/1	1/1	1/1	1/1	1.8/1.3	1.5/1	1/1			1.8/1	1.5/1	1.5/1	1.5/1	1.5/1	1.5/1	
te [O,RO*PI]																			



PROGRAMAS

```

10 REM      ! ***** PROGRAMA:  CURVAS1 *****  !
20 REM
30 REM      ! ESTE PROGRAMA TRAZA LAS CURVAS QUE FIGURAN EN LA
40 REM      ! RUTINA DE CALCULO.EN ELLA ESTAN LOS FACTORES DE ESCALA
50 REM      ! SI NO DESEA DIBUJAR LOS EJES PONGA: 205 GOTO 220
60 DISP "R";
70 INPUT R
80 DISP "F";
90 INPUT F
100 DISP "D";
110 INPUT D
120 DISP "L";
130 INPUT L
140 DISP "J";
150 INPUT J
160 PLOTTER IS 305
170 GRAPHALL
180 LIMIT 5,270,5,190
190 FRAME
200 SCALE -13.25,13.25,-9.5,9.5
210 AXES 1,1,0,0,10,5,2
220 P=.1 ! ELECCION DEL STEP
230 T=0
240 GOSUB 380
250 MOVE X,Y
260 FOR T=0 TO RO*PI STEP P ! SI NO SE MODIFICA EN LA RUTINA,RO TOMA VALOR 2

```

```

270 GOSUB 380
280 DRAW X,Y
290 NEXT T
300 DRAW X,Y
310 MOVE -13,-7 @ LABEL "X=R*COS(t)+F*COS(D*t)"
320 MOVE -13,-8.5 @ LABEL "Y=R*SEN(t)+J*SEN(L*t)"
330 MOVE 2,-7 @ LABEL "R=";R;" F=";F;" D=";D;" L=";L;" J=";J
340 MOVE 2,-8 @ LABEL "FACTORES DE ESCALA..(X/Y)=";A;" / ";B
350 MOVE 2,-9 @ LABEL "ANGULO RO=";RO
360 ALPHALL
370 END
380 REM      !RUTINA DE CALCULO!
390 X=R*COS (T)+F*COS (D*T)
400 Y=R*SIN (T)+J*SIN (L*T)
410 A=1 ! FACTOR DE ESCALA EN OX
420 B=1 ! FACTOR DE ESCALA EN OY
430 X=X*A
440 Y=Y*B
450 RO=2 ! VARIACION DE T=RO*PI
460 RETURN

```

```

10 REM      ! ***** PROGRAMA:  CURVAS2 *****
20 !
30 !
40 !      ESTE PROGRAMA UNE LOS PUNTOS DE LA CURVA QUE
50 !      FIGURA EN LA RUTINA CON UN PUNTO FIJO QUE,SINO SE DEFINE,
60 !      SE TOMA EL ORIGEN DE COORDENADAS.
70 !      UTILICE PAPEL VEGETAL EN EL PLOTTER
80 !      SUPRIMIR LA SENTENCIA 410,SI SE DESEA QUE FIGUREN LAS CARACTERISTICAS
90 DISP "R";
100 INPUT R
110 DISP "F";
120 INPUT F
130 DISP "D";
140 INPUT D
150 DISP "L";
160 INPUT L
170 DISP "J";
180 INPUT J
190 PLOTTER IS 305
200 GRAPHALL
210 LIMIT 5,270,5,190
220 FRAME
230 SCALE -13.25,13.25,-9.5,9.5
240 GOTO 260 ! SUPRIMIR SI SE QUIERE DIBUJAR LOS EJES
250 AXES 1,1,0,0,10,5,2
260 P=.01 ! ELECCION DEL STEP
270 XX=0 @ YY=0 ! ***** PUNTO DE FUGA *****
290 RO=2 ! VARIACION DE T=RO*PI
300 MOVE XX,YY
310 H=0
320 FOR T=0 TO RO*PI STEP P
330 IF H=1 THEN 350
340 MOVE XX,YY
350 GOSUB 490
360 DRAW X,Y
370 NEXT T
380 DRAW X,Y
390 H=H+1
400 IF H=1 THEN 320
410 GOTO 470
420 MOVE -13,-7 @ LABEL "X=R*COS(t)+F*COS(D*t)"
430 MOVE -13,-8.5 @ LABEL "Y=R*SEN(t)+J*SEN(L*t)"
440 MOVE 2,-7 @ LABEL "R=";R;" F=";F;" D=";D;" L=";L;" J=";J
450 MOVE 2,-8 @ LABEL "FACTORES DE ESCALA..(X/Y)=";A;" / ";B
460 MOVE 2,-9 @ LABEL "ANGULO RO=";RO
470 ALPHALL
480 END
490 !      RUTINA DE CALCULO
500 X=R*COS (T)+F*COS (D*T)
510 Y=R*SIN (T)+J*SIN (L*T)
520 A=1 ! FACTOR DE ESCALA EN OX
530 B=1 ! FACTOR DE ESCALA EN OY
540 X=X*A
550 Y=Y*B
560 RETURN

```

